
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.3.1.14a which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2012-03_D12wk36r1’.
The test case can be demonstrated to run with one LTE UE (see section 5) on two different test platforms. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_3_1_14a
Test Group:
8_3\RRC_Measurement.ttcn

ATS Version:
iwd-EUTRA-B2012-03_D12wk36r1
System Simulators used:
AT4 wireless T4110 (Agilent T4110S)  +
   Rohde & Schwarz CMW500 
UE used:
Huawei Mobile WiFi
Verification Status:
PASS


4. Corrections required for test case 8.3.1.14a
This section describes the TTCN changes required to make test case 8.3.1.14a run correctly with a LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2012-03_D12wk36r1’ release.

Change 1 – Corrections to template ‘cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4()’
	Template name
	cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4()

	Reason for change
	The template definition did not perfectly match the requirements of TS 36.523-1 Tables 8.3.1.14a.3.3-2 and 8.3.1.14a.3.3-14 

	Summary of change
	Template modified so that it can be used in test steps 1 and 8 passing the values like parameters “measObjectID[]”. Parameters “p_MeasObjectId_f1” and “p_MeasObjectId_f6” defined.

	TTCN module
	8_3\EUTRA_Measurement_Specific_Templates.ttcn

	MCC160 Comment
	


Before change

…
    //8.3.1.14a, Step1
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4(RRC_TransactionIdentifier p_RRC_TI,
                                                                                               ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq_1,
                                                                                               AllowedMeasBandwidth p_MeasurementBandwidth_1,
                                                                                               ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq_6,
                                                                                               AllowedMeasBandwidth p_MeasurementBandwidth_6 ) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    cs_MeasObjectToAddModList_2Entries(tsc_IdMeasObject_f1,
                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq_1, p_MeasurementBandwidth_1, omit, omit, omit ),
                                                                                                       tsc_IdMeasObject_f6,
                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq_6, p_MeasurementBandwidth_6, omit, omit, omit ) ),
                                                                    omit,
                                                                    cs_ReportConfigToAddModList_2Entries(tsc_IdReportConfig_A2,
                                                                                                         cs_ReportConfig_reportConfigEUTRA_eventA2_RSRP( -83, 6, ms320, rsrp, both, 1, ms1024, r1 ),
                                                                                                         tsc_IdReportConfig_A3,
                                                                                                         cs_ReportConfig_reportConfigEUTRA_eventA3( 0, false, 0, ms640, rsrp, sameAsTriggerQuantity, 1, ms1024, r1 ) ),
                                                                    omit,
                                                                    cs_MeasIdToAddModList_2Entries(1, tsc_IdMeasObject_f6, tsc_IdReportConfig_A2,
                                                                                                   2, tsc_IdMeasObject_f1, tsc_IdReportConfig_A3 ),
                                                                    cs_QuantityConfig_Def,
                                                                    cs_508_MeasGapConfig_GP1 ) );
    //------------------------------------------------
…
After change

…
    //8.3.1.14a, Step1
  template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4(RRC_TransactionIdentifier p_RRC_TI,
                                                                                               ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq_1,
                                                                                               AllowedMeasBandwidth p_MeasurementBandwidth_1,
                                                                                               ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq_6,
                                                                                               AllowedMeasBandwidth p_MeasurementBandwidth_6,
                                                                                               MeasObjectId p_MeasObjectId_f1,
                                                                                               MeasObjectId p_MeasObjectId_f6 ) :=
    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,
                                                      cs_MeasConfig(omit,
                                                                    cs_MeasObjectToAddModList_2Entries(tsc_IdMeasObject_f1,
                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq_1, p_MeasurementBandwidth_1, omit, omit, omit ),
                                                                                                       tsc_IdMeasObject_f6,
                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq_6, p_MeasurementBandwidth_6, omit, omit, omit ) ),
                                                                    omit,
                                                                    cs_ReportConfigToAddModList_2Entries(tsc_IdReportConfig_A2,
                                                                                                         cs_ReportConfig_reportConfigEUTRA_eventA2_RSRP( -83, 6, ms320, rsrp, both, 1, ms1024, r1 ),
                                                                                                         tsc_IdReportConfig_A3,
                                                                                                         cs_ReportConfig_reportConfigEUTRA_eventA3( 0, false, 0, ms640, rsrp, sameAsTriggerQuantity, 1, ms1024, r1 ) ),
                                                                    omit,
                                                                    cs_MeasIdToAddModList_2Entries(1, p_MeasObjectId_f1, tsc_IdReportConfig_A2,
                                                                                                   2, p_MeasObjectId_f6, tsc_IdReportConfig_A3 ),
                                                                    cs_QuantityConfig_Def,
                                                                    cs_508_MeasGapConfig_GP1 ) );
    //------------------------------------------------
…
Change 2 – Correction to import part of TTCN module “RRC_Measurement.ttcn”
	Module part 
	Import part

	Reason for change
	Import from TTCN module “RRC_handover.ttcn” was missing as prerequisite to use function “f_EUTRA_ConnectedMode_TAU_EPLMN()” ins steps 8a3-8a5. 
(… see change 3 subchange 7 below …)
 

	Summary of change
	“Import from RRC_Handover” all added.

	TTCN module
	8_3\RRC_Measurement.ttcn

	MCC160 Comment
	


Before change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2012, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: LTE_IWD_12wk36_r1
// $Date: 2012-09-06 14:14:27 +0200 (Thu, 06 Sep 2012) $
// $Rev: 7338 $
/******************************************************************************/
module RRC_Measurement
{
/////////////////
// Import Part //
/////////////////
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_Component all;
  import from Parameters all;
  import from UpperTesterFunctions all;
  import from CommonDefs all;
  import from EUTRA_CommonDefs all;
  import from CommonIratDefs all;
  import from EUTRA_IdleMode all;
  import from EUTRA_RRC_Templates all;
  import from EUTRA_SRB_Templates all;
  import from EUTRA_AspCommon_Templates all;
  import from EUTRA_CellInfo all;
  import from EUTRA_CommonProcedures all;
  import from EUTRA_ConfigurationSteps all;
  import from EUTRA_Measurement_Templates all;
  import from EUTRA_Measurement_Specific_Templates all;
  import from EUTRA_Timing all;
  import from EUTRA_RRCSteps all;
  import from EUTRA_CellCfg_Templates all;
  import from EUTRA_SecuritySteps all;
  import from EUTRA_SecurityFunctions all;
  import from EUTRA_ASP_TypeDefs all;
  import from EUTRA_ASP_SrbDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from RRC_Measurement_Functions all;
  import from EUTRA_NASSteps all;
  import from EUTRA_AuxiliaryFunctions all;
  import from EUTRA_BandDependentParam all;
  import from EPS_NAS_Templates all;
  import from EPS_NAS_TypeDefs all;
  import from EPS_NAS_Constants all;
  ////////////////////
  // Functions Part //
  ////////////////////
…

After change

/******************************************************************************/
// @copyright   Copyright Notification
//   No part may be reproduced except as authorized by written permission.
//   The copyright and the foregoing restriction extend to reproduction in all media.
//   Trademark 2012, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TTA, TTC).
//   All rights reserved.
// @version: LTE_IWD_12wk36_r1
// $Date: 2012-09-06 14:14:27 +0200 (Thu, 06 Sep 2012) $
// $Rev: 7338 $
/******************************************************************************/
module RRC_Measurement
{
/////////////////
// Import Part //
/////////////////
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from EUTRA_Component all;
  import from Parameters all;
  import from UpperTesterFunctions all;
  import from CommonDefs all;
  import from EUTRA_CommonDefs all;
  import from CommonIratDefs all;
  import from EUTRA_IdleMode all;
  import from EUTRA_RRC_Templates all;
  import from EUTRA_SRB_Templates all;
  import from EUTRA_AspCommon_Templates all;
  import from EUTRA_CellInfo all;
  import from EUTRA_CommonProcedures all;
  import from EUTRA_ConfigurationSteps all;
  import from EUTRA_Measurement_Templates all;
  import from EUTRA_Measurement_Specific_Templates all;
  import from EUTRA_Timing all;
  import from EUTRA_RRCSteps all;
  import from EUTRA_CellCfg_Templates all;
  import from EUTRA_SecuritySteps all;
  import from EUTRA_SecurityFunctions all;
  import from EUTRA_ASP_TypeDefs all;
  import from EUTRA_ASP_SrbDefs all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from RRC_Measurement_Functions all;
  import from EUTRA_NASSteps all;
  import from EUTRA_AuxiliaryFunctions all;
  import from EUTRA_BandDependentParam all;
  import from EPS_NAS_Templates all;
  import from EPS_NAS_TypeDefs all;
  import from EPS_NAS_Constants all;
  import from RRC_Handover all;
  ////////////////////
  // Functions Part //
  ////////////////////
…  
Change 3 – Corrections to function ‘f_TC_8_3_1_14a_EUTRA()’
	Function name
	f_TC_8_3_1_14a_EUTRA()

	Reason for change
	1. Incorrect PLMN identity mapping for cell 1 and 28.


2. SIB5 was not configured for intermode.


3. Power levels of the serving cell were no set at the beginning of the TC.

4. Template which did not match the RSRQ measurement requirements on cell 28 was used in steps 7 and 13 (step 14 in TTCN comments below). See TS 36.523-1 Table 8.3.1.14a.3.2-2.

5. Incorrect cell ID was used in the step that brings the UE to initial state with Equivalent PLMN2 prior to start of the Test Body as well as in subsequent step 8b2 and in the postamble.

6. Template used when calling function “f_EUTRA_RRCConnectionReconfiguration_MeasConfig()” in steps 1 – 2 did not match exactly to the Prose requiremennts (see change 1 avove).


7. As per TS 36.523-1 Table 8.3.1.14a.3.2-2 (steps 8a3 to 8a5) the SS should respond to  TAU REQUEST on cell 28 (which belongs to the EPLMN) from a UE supporting Feature Group 30  with TAU ACCEPT indicating PLMN1 as EPLMN. The latter was forgotten in the TTCN implementation.


8. As per step 8a1 of the main procedure sequence (TS 36.523-1 Table 8.3.1.14a.3.2-2) the UE should have been ordered to perform inter-band handover to cell 28 and to activate the measurement gaps.

	Summary of change
	1. Corrected the PLMN identities of cell 1 and 28 corrected as per TS 36.523-1 Table 6.0.1-1 (… which is to be used if cell / PLMN mapping is not otherwise specified in the respective TC).


2. Intermode configuration for SIB5 added prior to creation of cells.


3. Set the power levels of cell 1 right after creation of the cells.


4. Replaced the incorrect measurement template in steps 7 and 13.


5. Replaced the incorrect cell IDs with the “eutra_Cell28”.


6. Template as defined in change 1 is used when calling function “f_EUTRA_RRCConnectionReconfiguration_MeasConfig()” in steps 1 - 2.


7. Function call to “f_EUTRA_ConnectedMode_TAU()” replaced with “f_EUTRA_ConnectedMode_TAU_EPLMN()”.See also change 2 above.


8. Function “f_EUTRA_508RRC_IntraLTE_HO_InterCell()” is called in steps 8a1 with measurement gap parameters configured).


	TTCN module
	8_3\RRC_Measurement.ttcn

	MCC160 Comment
	f_EUTRA_508RRC_IntraLTE_HO_InterCell


Before change
…  
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.3.1.14a
   * @status
   * @status
   */
  function f_TC_8_3_1_14a_EUTRA() runs on EUTRA_PTC
  { /* Measurement configuration control and reporting / Intra E-UTRAN measurements / Two simultaneous events A2 and A3 (inter-band measurements) / Between FDD and TDD */
    var template (value) CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) CellPowerList_Type v_CellPowerList_AtT4;
    var Frequency_IE_Type v_Frequency_IE_f1;
    var Frequency_IE_Type v_Frequency_IE_f6;
    var ChannelBandwidthDependency_Type v_ChBandDependency_f1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_f6;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var PhysCellId v_PhysicalCellIdentity_Cell1;
    var PhysCellId v_PhysicalCellIdentity_Cell28;
    var float v_TimerValue ;
    v_CellPowerList_AtT0 := {
      cs_CellPower (eutra_Cell1, -75),
      cs_CellPower (eutra_Cell28, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    v_CellPowerList_AtT1 := {
      cs_CellPower (eutra_Cell1, -93),
      cs_CellPower (eutra_Cell28, -105)
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower (eutra_Cell28, -73)
    };
    v_CellPowerList_AtT3 := {
      cs_CellPower (eutra_Cell1, -105),
      cs_CellPower (eutra_Cell28, -93)
    };
    v_CellPowerList_AtT4 := {
      cs_CellPower (eutra_Cell1, -73),
      cs_CellPower (eutra_Cell28, tsc_ServingCellRS_EPRE)
    };
    //@sic R5-122102 sic@
    f_EUTRA_Init (c3);
    v_TimerValue := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 10.0);
    v_PhysicalCellIdentity_Cell1:= f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);
    v_PhysicalCellIdentity_Cell28 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell28);
    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs( eutra_Cell1 );
    v_Frequency_IE_f6 := f_EUTRA_CellInfo_GetFrequencyIEs( eutra_Cell28 );
    v_ChBandDependency_f1 := f_EUTRA_BandDependentParam( v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth );
    v_ChBandDependency_f6 := f_EUTRA_BandDependentParam( v_Frequency_IE_f6.DL_ChBandwidth, v_Frequency_IE_f6.UL_ChBandwidth );
    //Set maximum cell powel level for Cells to be used in creation
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -73);
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell28, -73);
    //@sic R5-122102 sic@
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell1,  cs_HPLMN_001_11 );  //PLMN1 according to 36.523-1 Table 6.0.1-1
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell28, cs_HPLMN_001_21 );  //PLMN2 according to 36.523-1 Table 6.0.1-1
    f_EUTRA_CellInfo_SetTAC (eutra_Cell1, tsc_IdleMode_TAC_Cell1);
    f_EUTRA_CellInfo_SetTAC (eutra_Cell28, tsc_IdleMode_TAC_Cell28);
    //Check if Cells configured correctly under FDD and TDD mode
    if (f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1)!= FDD) {
        FatalError (__FILE__, __LINE__, "Cell1 should be configured as a FDD cell");
    }
    if (f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell28)!= TDD) {
        FatalError (__FILE__, __LINE__, "Cell28 should be configured as a TDD cell");
    }
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_CellConfig_Def (eutra_Cell28);
    //@sic R5-122102 sic@
    //Bring UE to inital state with Equivalent PLMN2
    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell1,
                                   PREAMBLE,
                                   cs_PLMN_List_1PLMN (f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell30))));
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);
    //------------------------------------------------
    //Start TestBody
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1-2" siclog@
    //SS transmits an RRCConnectionReconfiguration message including measConfig to setup intra LTE measurement and reporting for event A2 and event A3 (inter-band measurement) on Cell 1
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4( v_RRC_TI,
                                                                                  v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f1.AllowedMeasBandwidth,
                                                                                  v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f6.AllowedMeasBandwidth ));
    //@siclog "Step 3" siclog@
    //Check: Does the UE transmit a MeasurementReport message within the next 10s?
    f_NoMeasurementReport(eutra_Cell1, v_TimerValue, cr_MeasurementReport_CheckAny);
    //@siclog "Step 4" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T1" in table 8.3.1.14a.3.2.-1..
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 5" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A2 with the measured RSRP value for Cell 1?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 5");
    //@siclog "Step 6" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T2" in table 8.3.1.14a.3.2.-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    //@siclog "Step 7" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for Cell 28?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport_Eutra_1Entry_both(2, v_PhysicalCellIdentity_Cell28)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 7");
    //@sic R5-122102 sic@
    //@siclog "Step 8a1-8a5,8b1-8b5" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-band handover to Cell 28 and to activate the measurement gaps.
    if (pc_FeatrGrp_30)
    {
    //@siclog "Step 8a1-8a2" siclog@
    //F pc_FeatrGrp_30 THEN the SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-band handover to Cell 28 and to activate the measurement gaps.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 28 to confirm the handover
    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1, eutra_Cell28 );
    //@siclog "Step 8a3-8a5" siclog@
    //The UE transmits a TRACKING AREA UPDATE REQUEST message on Cell 28
    //SS responds with a TRACKING AREA UPDATE ACCEPT message.
    //NOTE: The TAU is accepted with PLMN1 listed as an Equivalent PLMN
    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.
    f_EUTRA_ConnectedMode_TAU ( eutra_Cell28 );
    }
    else
    {
     //@siclog "Step 8b1" siclog@
     //ELSE IF NOT pc_FeatrGrp_30 THEN the SS transmits an RRCConnectionRelease message  (IE redirectedCarrierInfo including eutra CarrierFreq of Cell 28) on Cell 1
     f_EUTRA_RRC_ConnectionRelease_Common ( eutra_Cell1,
                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,
                                                                     other,
                                                                     cs_RedirectionInfo_Eutra ( v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq ),                                                                 omit ) );
    //@siclog "Step 8b2" siclog@
    //The generic test procedure described in TS 36.508 subclause 6.4.2.7 is performed on Cell 28
    //NOTE: The TAU is accepted with PLMN1 listed as an Equivalent PLMN
    f_EUTRA_TrackingAreaUpdate_WithEplmn (eutra_Cell30,
                                          tsc_RRC_TI_Def,
                                          cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId(f_EUTRA_CellInfo_GetGutiPLMN(eutra_Cell1))));
    //@siclog "Step 8b3" siclog@
    //Generic test procedure described in TS 36.508 subclause 4.5.3.3.is executed
    f_EUTRA_RbEst_Def ( eutra_Cell28 );
    //@siclog "Step 8b4-8b5" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-band measurement and reporting for event A2 and event A3 on Cell 28.
    //The UE transmits an RRCConnectionReconfigurationComplete message.
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell28,
                                                          cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4( v_RRC_TI,
                                                                                  v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f1.AllowedMeasBandwidth,
                                                                                  v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f6.AllowedMeasBandwidth ));
      }
    //@siclog "Step 10" siclog@
    //Check: Does the UE transmit a MeasurementReport message within the next 10s?
    f_NoMeasurementReport(eutra_Cell28, v_TimerValue, cr_MeasurementReport_CheckAny);
    //@siclog "Step 11" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T3" in table 8.3.1.14a.3.2.-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);
    //@siclog "Step 12" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A2 with the measured RSRP value for Cell 28?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell28,
                                    cr_MeasurementReport(1, ?, ?, omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 12");
    //@siclog "Step 13" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T4" in table 8.3.1.12a.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT4);
    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for Cell 1?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell28,
                                    cr_MeasurementReport_Eutra_1Entry_both(1, v_PhysicalCellIdentity_Cell1)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 14");
    //End TestBody
    //------------------------------------------------
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  } //end of testcase 8.3.1.14a
  //================================================================
…  

After change

…  
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.3.1.14a
   * @status
   * @status
   */
  function f_TC_8_3_1_14a_EUTRA() runs on EUTRA_PTC
  { /* Measurement configuration control and reporting / Intra E-UTRAN measurements / Two simultaneous events A2 and A3 (inter-band measurements) / Between FDD and TDD */
    var template (value) CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) CellPowerList_Type v_CellPowerList_AtT4;
    var Frequency_IE_Type v_Frequency_IE_f1;
    var Frequency_IE_Type v_Frequency_IE_f6;
    var ChannelBandwidthDependency_Type v_ChBandDependency_f1;
    var ChannelBandwidthDependency_Type v_ChBandDependency_f6;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var PhysCellId v_PhysicalCellIdentity_Cell1;
    var PhysCellId v_PhysicalCellIdentity_Cell28;
    var float v_TimerValue ;
    v_CellPowerList_AtT0 := {
      cs_CellPower (eutra_Cell1, -75),
      cs_CellPower (eutra_Cell28, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };
    v_CellPowerList_AtT1 := {
      cs_CellPower (eutra_Cell1, -93),
      cs_CellPower (eutra_Cell28, -105)
    };
    v_CellPowerList_AtT2 := {
      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower (eutra_Cell28, -73)
    };
    v_CellPowerList_AtT3 := {
      cs_CellPower (eutra_Cell1, -105),
      cs_CellPower (eutra_Cell28, -93)
    };
    v_CellPowerList_AtT4 := {
      cs_CellPower (eutra_Cell1, -73),
      cs_CellPower (eutra_Cell28, tsc_ServingCellRS_EPRE)
    };
    //@sic R5-122102 sic@
    f_EUTRA_Init (c3);
    v_TimerValue := f_EUTRA_SetTimerToleranceMax(eutra_Cell1, nonProtocolTimer, 10.0);
    v_PhysicalCellIdentity_Cell1:= f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);
    v_PhysicalCellIdentity_Cell28 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell28);
    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs( eutra_Cell1 );
    v_Frequency_IE_f6 := f_EUTRA_CellInfo_GetFrequencyIEs( eutra_Cell28 );
    v_ChBandDependency_f1 := f_EUTRA_BandDependentParam( v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth );
    v_ChBandDependency_f6 := f_EUTRA_BandDependentParam( v_Frequency_IE_f6.DL_ChBandwidth, v_Frequency_IE_f6.UL_ChBandwidth );
    //Set maximum cell powel level for Cells to be used in creation
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -73);
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell28, -73);
    //@sic R5-122102 sic@
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell1,  cs_HPLMN_Def );  //PLMN1 according to 36.523-1 Table 6.0.1-1
    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell28, cs_HPLMN_002_11 );  //PLMN2 according to 36.523-1 Table 6.0.1-1
    f_EUTRA_CellInfo_SetTAC (eutra_Cell1, tsc_IdleMode_TAC_Cell1);
    f_EUTRA_CellInfo_SetTAC (eutra_Cell28, tsc_IdleMode_TAC_Cell28);
    //Check if Cells configured correctly under FDD and TDD mode
    if (f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1)!= FDD) {
        FatalError (__FILE__, __LINE__, "Cell1 should be configured as a FDD cell");
    }
    if (f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell28)!= TDD) {
        FatalError (__FILE__, __LINE__, "Cell28 should be configured as a TDD cell");
    }
    //c3 – set SIB5 according to intermode values
    f_EUTRA_SetSIB5_interMode (eutra_Cell1);
    f_EUTRA_SetSIB5_interMode (eutra_Cell28);
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_CellConfig_Def (eutra_Cell28);
    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);
    //Switch on UE
    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);
    //@sic R5-122102 sic@
    //Bring UE to inital state with Equivalent PLMN2
    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell1,
                                   PREAMBLE,
                                   cs_PLMN_List_1PLMN (f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell28))));
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    //------------------------------------------------
    //Start TestBody
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1-2" siclog@
    //SS transmits an RRCConnectionReconfiguration message including measConfig to setup intra LTE measurement and reporting for event A2 and event A3 (inter-band measurement) on Cell 1
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,
                                                    cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4( v_RRC_TI,
                                                                                  v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f1.AllowedMeasBandwidth,
                                                                                  v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f6.AllowedMeasBandwidth,
                                                                                  tsc_IdMeasObject_f1,
                                                                                  tsc_IdMeasObject_f6 ));
    //@siclog "Step 3" siclog@
    //Check: Does the UE transmit a MeasurementReport message within the next 10s?
    f_NoMeasurementReport(eutra_Cell1, v_TimerValue, cr_MeasurementReport_CheckAny);
    //@siclog "Step 4" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T1" in table 8.3.1.14a.3.2.-1..
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);
    //@siclog "Step 5" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A2 with the measured RSRP value for Cell 1?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport(1, ?, ?, omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 5");
    //@siclog "Step 6" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T2" in table 8.3.1.14a.3.2.-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    //@siclog "Step 7" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for Cell 28?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_MeasurementReport_Eutra_1Entry_rsrp(2, v_PhysicalCellIdentity_Cell28)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 7");
    //@sic R5-122102 sic@
    //@siclog "Step 8a1-8a5,8b1-8b5" siclog@
    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-band handover to Cell 28 and to activate the measurement gaps.
    if (pc_FeatrGrp_30)
    {
    //@siclog "Step 8a1-8a2" siclog@
    //F pc_FeatrGrp_30 THEN the SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform inter-band handover to Cell 28 and to activate the measurement gaps.
    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 28 to confirm the handover
    f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1, eutra_Cell28, -, -,
                                           cs_MeasConfig(omit,
                                                         omit,
                                                         omit,
                                                         omit,
                                                         omit,
                                                         omit,
                                                         cs_QuantityConfig_Def,
                                                         cs_508_MeasGapConfig_GP1));
    //@siclog "Step 8a3-8a5" siclog@
    //The UE transmits a TRACKING AREA UPDATE REQUEST message on Cell 28
    //SS responds with a TRACKING AREA UPDATE ACCEPT message.
    //NOTE: The TAU is accepted with PLMN1 listed as an Equivalent PLMN
    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.
    f_EUTRA_ConnectedMode_TAU_EPLMN ( eutra_Cell28, cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId(f_EUTRA_CellInfo_GetGutiPLMN(eutra_Cell1))) );
    }
    else
    {
     //@siclog "Step 8b1" siclog@
     //ELSE IF NOT pc_FeatrGrp_30 THEN the SS transmits an RRCConnectionRelease message  (IE redirectedCarrierInfo including eutra CarrierFreq of Cell 28) on Cell 1
     f_EUTRA_RRC_ConnectionRelease_Common ( eutra_Cell1,
                                           cs_RRCConnectionRelease ( tsc_RRC_TI_Def,
                                                                     other,
                                                                     cs_RedirectionInfo_Eutra ( v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq ),                                                                 omit ) );
    //@siclog "Step 8b2" siclog@
    //The generic test procedure described in TS 36.508 subclause 6.4.2.7 is performed on Cell 28
    //NOTE: The TAU is accepted with PLMN1 listed as an Equivalent PLMN
    f_EUTRA_TrackingAreaUpdate_WithEplmn (eutra_Cell28,
                                          tsc_RRC_TI_Def,
                                          cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId(f_EUTRA_CellInfo_GetGutiPLMN(eutra_Cell1))));
    //@siclog "Step 8b3" siclog@
    //Generic test procedure described in TS 36.508 subclause 4.5.3.3.is executed
    f_EUTRA_RbEst_Def ( eutra_Cell28 );
    //@siclog "Step 8b4-8b5" siclog@
    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-band measurement and reporting for event A2 and event A3 on Cell 28.
    //The UE transmits an RRCConnectionReconfigurationComplete message.
    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell28,
                                                          cs_RRCConnectionReconfiguration_8_3_1_14a_Step1_and_Step8b4( v_RRC_TI,
                                                                                  v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f1.AllowedMeasBandwidth,
                                                                                  v_Frequency_IE_f6.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                  v_ChBandDependency_f6.AllowedMeasBandwidth,
                                                                                  tsc_IdMeasObject_f6,
                                                                                  tsc_IdMeasObject_f1 ));
      }
    //@siclog "Step 10" siclog@
    //Check: Does the UE transmit a MeasurementReport message within the next 10s?
    f_NoMeasurementReport(eutra_Cell28, v_TimerValue, cr_MeasurementReport_CheckAny);
    //@siclog "Step 11" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T3" in table 8.3.1.14a.3.2.-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);
    //@siclog "Step 12" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A2 with the measured RSRP value for Cell 28?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell28,
                                    cr_MeasurementReport(1, ?, ?, omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 12");
    //@siclog "Step 13" siclog@
    //SS re-adjusts the cell-specific reference signal level according to row "T4" in table 8.3.1.12a.3.2-1.
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT4);
    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for Cell 1?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell28,
                                    cr_MeasurementReport_Eutra_1Entry_rsrp(2, v_PhysicalCellIdentity_Cell1)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.3.1.14a Step 14");
    //End TestBody
    //------------------------------------------------
    f_EUTRA_TestBody_Set(false);
    f_EUTRA_Postamble(eutra_Cell28, E2_CONNECTED);
  } //end of testcase 8.3.1.14a
  //================================================================
…  
5. Execution Log Files

5.1 Huawei Mobile WiFi UE 

The Huawei Mobile WiFi UE passed this test case on AT4 wireless T4110 (= Agilent T4110S) and 
R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 4 and LTE TDD band 41. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files from Agilent Technologies:
8_3_1_14a folder
In the log files (in html format) the complete test case execution can be seen. PICS/PIXIT settings are captured at the beginning of the Traces.html file. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.
· Test Case Execution log files from R&S: 
tc_8_3_1_14a_Huawei_FDD_band_4_TDD_band_41.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

